
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 9.2 which is part of the ‘IMS_34229_IWD_13wk08’ test suite in ATS ‘iwd-EUTRA-B2012-03_D13wk08’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.
2. Table of Contents
101 Jan – 31 Dec 2013


21. Overview


22. Table of Contents


23. Verification Test Summary


34. Corrections required


3Change 1 - f_IMS_RegisterRequest_MessageHeaderRX_Initial


4Change 2 - f_TC_9_2_IMS


65. Execution Log Files


65.1
STE M7400 UE


66. References




3. Verification Test Summary

Test Case:
9.2
Test Group:
IMS Authentication
ATS Version:
iwd-EUTRA-B2012-03_D13wk08
System Simulator used:
Rohde & Schwarz Multi-RAT Protocol Tester
UE used:
Intel XMM7160 chip
Verification Status:
PASS
4. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk08’ release.
Change 1 - f_IMS_RegisterRequest_MessageHeaderRX_Initial
	Function name
	f_IMS_RegisterRequest_MessageHeaderRX_Initial

	Reason for change
	Allow rport in Register requests

	Summary of change
	See R5s130187 for definition of cr_ViaDef2 (resp the alternative implementation in form of new function f_ViaParamsRX proposed by MCC160 in their comments to R5s130187, Change 6)

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration

	MCC160 Comment
	


Before change
function f_IMS_RegisterRequest_MessageHeaderRX_Initial(boolean p_CondA3 := false) runs on IMS_PTC return template (present) MessageHeader

  { /* initial REGISTER (condition A1 or A3) acc. to 34.229-1 Annex A.1.1

       p_CondA3  .. condition A3 acc. to 34.229-1 Annex A.1.1 (REGISTER for the case UE supports GIBA)

       NOTE: several checks are done explicitly in the respective check functions applied on the receive messages (the same for sigcomp) */

    var template (present) SipUrl v_SipUrlPublicUserId := ?;   /* Acc. to 24.229 cl. 5.1.1.2.1 for initial registration the UE shall populate

                                                                  'To' and 'From' with the public user identity to be registered;

                                                                  Normally this is one of the IMPUs stored on the ISIM;

                                                                  The public user identity is explicitly checked after matching the REGISTER message */

    var template (present) integer v_CseqValue := ?;

    var template (present) MessageHeader v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(p_CondA3, v_CseqValue);

    var template (present) Via v_Via := cr_ViaInitialReg;

    if (p_CondA3) { v_Via := cr_ViaDef; }     // "rport" is not mandatory for A3        @sic R5s120858 change 2 sic@

    v_MessageHeader_Register.via         := v_Via;

    v_MessageHeader_Register.route       := cr_RouteRegDef(px_IMS_Pcscf, *) ifpresent;

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);  // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);        // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUrl_HostPort(?, *));  // checked in fl_IMS_ContactAddress_CheckParams, f_IMS_A_1_1_MessageHeader_CommonChecks

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *) ifpresent;    // optional for A1, A3; cellId is not checked

    if (not p_CondA3) {  // => 34.229-1 A.1.1 condition A1

      v_MessageHeader_Register.securityVerify := omit;

      v_MessageHeader_Register.authorization := cr_AuthorizationInitial(px_IMS_Private_UserId, px_IMS_HomeDomainName);

    }

    return v_MessageHeader_Register;

  }
After change
function f_IMS_RegisterRequest_MessageHeaderRX_Initial(boolean p_CondA3 := false) runs on IMS_PTC return template (present) MessageHeader

  { /* initial REGISTER (condition A1 or A3) acc. to 34.229-1 Annex A.1.1

       p_CondA3  .. condition A3 acc. to 34.229-1 Annex A.1.1 (REGISTER for the case UE supports GIBA)

       NOTE: several checks are done explicitly in the respective check functions applied on the receive messages (the same for sigcomp) */

    var template (present) SipUrl v_SipUrlPublicUserId := ?;   /* Acc. to 24.229 cl. 5.1.1.2.1 for initial registration the UE shall populate

                                                                  'To' and 'From' with the public user identity to be registered;

                                                                  Normally this is one of the IMPUs stored on the ISIM;

                                                                  The public user identity is explicitly checked after matching the REGISTER message */

    var template (present) integer v_CseqValue := ?;

    var template (present) MessageHeader v_MessageHeader_Register := fl_IMS_RegisterRequest_MessageHeaderRX_Base(p_CondA3, v_CseqValue);

    var template (present) Via v_Via := cr_ViaDef2; 
    if (p_CondA3) { v_Via := cr_ViaDef; }     // "rport" is not mandatory for A3        @sic R5s120858 change 2 sic@

    v_MessageHeader_Register.via         := v_Via;

    v_MessageHeader_Register.route       := cr_RouteRegDef(px_IMS_Pcscf, *) ifpresent;

    v_MessageHeader_Register.fromField   := cr_FromWithTag(v_SipUrlPublicUserId);  // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.toField     := cr_ToDef(v_SipUrlPublicUserId);        // checked in f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity

    v_MessageHeader_Register.contact     := cr_Contact(cr_SipUrl_HostPort(?, *));  // checked in fl_IMS_ContactAddress_CheckParams, f_IMS_A_1_1_MessageHeader_CommonChecks

    v_MessageHeader_Register.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *) ifpresent;    // optional for A1, A3; cellId is not checked

    if (not p_CondA3) {  // => 34.229-1 A.1.1 condition A1

      v_MessageHeader_Register.securityVerify := omit;

      v_MessageHeader_Register.authorization := cr_AuthorizationInitial(px_IMS_Private_UserId, px_IMS_HomeDomainName);

    }

    return v_MessageHeader_Register;

  }
Change 2 - f_TC_9_2_IMS
	Function name
	 f_TC_9_2_IMS

	Reason for change
	1. Security is not initialized early enough
2. Routing info is out of date

	Summary of change
	1. Add call to f_IMS_Register_SecurityInit() before step 2
2. Feed f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo)  rather than v_RoutingInfo_DL into template cas_IMS_DATA_RSP

	TTCN module
	 IMS_34229/IMS_CC_AuthenticationTestcases

	MCC160 Comment
	


Before change

function f_TC_9_2_IMS() runs on IMS_PTC
  {

    var charstring v_InvalidNonce := fl_GenerateInvalidNonce(sqnFailure);

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var REGISTER_Request v_RegisterReq;

    var charstring v_CallId;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Step 1

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    v_CallId := v_RegisterReq.msgHeader.callId.callid;

    // Step 2

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq, v_InvalidNonce))));

    // Step 3

    v_IMS_DATA_REQ := f_IMS_REGISTER_RequestWithSpecificAuth(v_CallId, cr_Authorization_9_2(v_InvalidNonce, tsc_IMS_Opaque));

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    f_IMS_Register_SecurityInit(v_RegisterReq);

    // Step 4

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));

    f_IMS_Register_SubscribeNotify(v_RegisterReq);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }
After change

function f_TC_9_2_IMS() runs on IMS_PTC
  {

    var charstring v_InvalidNonce := fl_GenerateInvalidNonce(sqnFailure);

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var REGISTER_Request v_RegisterReq;

    var charstring v_CallId;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL);

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    f_IMS_TestBody_Set(true);

    // Step 1

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    f_IMS_Register_SecurityInit(v_RegisterReq); 
    v_CallId := v_RegisterReq.msgHeader.callId.callid;

    // Step 2

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq, v_InvalidNonce))));

    // Step 3

    v_IMS_DATA_REQ := f_IMS_REGISTER_RequestWithSpecificAuth(v_CallId, cr_Authorization_9_2(v_InvalidNonce, tsc_IMS_Opaque));

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    f_IMS_Register_SecurityInit(v_RegisterReq);

    // Step 4

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // Step 5

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest();

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    // Step 6

    // CR against 34.229 needed as we should proceed with Step 7 instead of Step 5 in table "Expected sequence", step 6 where we talk about where to proceed in Annex C.2: 

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq))));
    f_IMS_Register_SubscribeNotify(v_RegisterReq);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_InitialRegistration);

  }
5. Execution Log Files

5.1  XMM7160 Intel chip
XMM7160 Intel chip passed this test case on the Rohde & Schwarz Multi-RAT Protocol Tester. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_IntelXMM7160.log:
In the log files (in txt format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6. References

[1] R5s130265: Supporting information for agreement of IMS test case 9.2 over LTE with 34.229-3 test model. This archive comprises txt format execution log file.

